

```
% Solutions for the problems from "TDDD08 - Tutorial 5"
```

```
% 1:
% A palindrome is the same thing reversed.
```

```
palindrome(L):-reverse(L,L).
```

```
% Two versions of reverse are presented in the handout on difference lists.
% Alternatively, it is provided by the library "lists"
```

```
% 2:
% thi(L,M): M is the list of every third element in L, starting from the 2nd one.
% thi1(L,M): M is the list of every third element in L, starting from the 1st one.
% thi3(L,M): M is the list of every third element in L, starting from the 3rd one.
```

```
thi([],[]).
thi([_|R],L):-thi1(R,L).
```

```
thi1([],[]).
thi1([H|R],[H|L]):-thi3(R,L).
```

```
thi3([],[]).
thi3([_|R],L):-thi(R,L).
```

```
% 3:
% Just following the definition, given the representation
% leaf(_) for a leaf and tree(L,R) for a node with two subtrees L,R.
```

```
symm(leaf(X),leaf(X)).
symm(tree(L1,R1),tree(L2,R2)):-
    symm(L1,R2),symm(R1,L2).
```

```
% 4:
x([],[],[]).
x([X|Xs],[X|Ys],Zs) :- x(Xs,Zs,Ys).
```

```
% For a list [a1,a2,...,an], let us call
% a1,a3,... its odd elements, and a2,a4,... its even elements.
```

```
% This is an informal description of the relation defined by the program
```

```
% x(X,Y,Z) - list X is split into lists Y,Z,
%             Y contains the odd elements of X,
%             Z contains the even elements of X.
```

```
% Because
% if Zs are the odd elements of Xs then they are the even elements of [X|Xs],
% if Ys are the even elements of Xs then they, together with X, are the odd
% elements of [X|Xs],
```

```
% A way to find out the relation was to construct some proof tree roots:
```

```
%   x([],[],[]).
%   x([X1],[X1],[]).
%   x([X2,X1],[X2],[X1]).
%   x([X3,X2,X1],[X3,X1],[X2]).
% (or to start constructing the least Herbrand model of the program).
```

```
% 5:
% A cycle of length 3 means three arcs (N1,N2), (N2,N3), (N3,N1).
% (Ni,Nj) is an arc if Nj is in the list Njs of nodes connected to Ni.
% The latter means that pair Ni-Njs is in the list representing the graph.
```

```
% three_cycle(G) - there is a cycle of length 3 in a graph G.
```

```
three_cycle(G):- arch(N1,N2,G), arch(N2,N3,G), arch(N3,N1,G).
```

```
% arc(N,M,G) - graph G has an arch (N,M).

arch(N,M,G) :- member(N-Ms,G),member(M,Ms).

% A test query:
%   arch(N,M, [a-[b,c], b-[c], c-[d,e], d-[f], e-[f,g], f-[g], g-[e]] ).

% 6:
% [] is a list of even length(0).
% Any other even length list is a list of odd length plus one extra element.
% [_] is a list of odd length, whatever the content.
% Any longer odd length list is a list of even length plus one extra element.

el([]).
el([_|L]):-ol(L).

ol([_|L]):-el(L).

% To be able to use the library predicate reverse:
:-use_module(library(lists)).
```